
==============================Simple_Eye_Charts_In_Spice========================================

1) Eye chart testing involves applying a random stream of one and zeros to a transmission line.
2) It is easy to generate random ones an zeros in spice.
3) A transmission line can be modeled as an series of LCR circuits.
4) A saw tooth waveform can be generated to plot the output signal in a eye chart format.
5) The bandwidth limitations on cables and transmission lines is not limitless.
6) Bandwidth spreads sharp signal over time since low pass filtering is another name for
   averaging signal over time.
7) The spreading of signal over time really starts to close the eye for a long enough cable. 

The transmission is modeled as a 20 stage LRC serial array with each stage having a bandwidth of
2GHz. Although resistance is inside this model, it looks like limited bandwidth by itself is
enough to start closing the eye over long enough cable. 

Limited bandwidth is just another name for low pass filtering which is another name for
averaging a sharp signal over time. Including more cycles in the eye charts can really
show how the sharp input signal is getting spread out over time as it goes through 
each stage. 

===================MacSpiceCode====================================================
Simple_Eye_Pattern
*             RIN     V0     V1     V2     V3     V4      V5    V6     V7     V8             V20
*                        ____   ____   ____   ____   ____   ____   ____   ____          ____
*        ___/\  /\  /\__|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|_|LRCG|...  ...|LRCG|__
*  VIN _|_    \/  \/    |____| |____| |____| |____| |____| |____| |____| |____|        |____|  |
*     /_  \               _|_    _|_    _|_    _|_    _|_    _|_    _|_    _|_           _|_   |
*    // \  \              ///    ///    ///    ///    ///    ///    ///    ///           ///   |
*    \   \//      ROUT                                                                         |
*     \___/    _/\  /\  /\_____________________________________________________________________|
*       |    _|_  \/  \/
*      _|_   ///
*      ///
*
*=========Need_A_voltage_Source_to_alter====================="
VIN       VIN    0    dc     0
Vref      VR     0    dc     0   PULSE( .25  .75 1p 1p  1p  1n 2n )
*=========Need_A_SawTooth====================="
VT        VT     0    dc     0   PWL    ( 0 0 1 1 ) 



BSAW      VSAW   0    V      =   2*atan(tan(6.2831*125meg*v(VT)-3.1416*.01))/3.1416
BSAW2     VSAW2  0    V      =   2*atan(tan(6.2831*125meg*v(VT)-3.1416*.1))/3.1416
*=========MODEL_Cable=========================
RIN       VIN    V0   50
CIN       V0     0    5p
XLRCG0    V0     V1   LRCG
XLRCG1    V1     V2   LRCG
XLRCG2    V2     V3   LRCG
XLRCG3    V3     V4   LRCG
XLRCG4    V4     V5   LRCG
XLRCG5    V5     V6   LRCG
XLRCG6    V6     V7   LRCG
XLRCG7    V7     V8   LRCG
XLRCG8    V8     V9   LRCG
XLRCG9    V9     V10  LRCG
XLRCG10   V10    V11  LRCG
XLRCG11   V11    V12  LRCG
XLRCG12   V12    V13  LRCG
XLRCG13   V13    V14  LRCG
XLRCG14   V14    V15  LRCG
XLRCG15   V15    V16  LRCG
XLRCG16   V16    V17  LRCG
XLRCG17   V17    V18  LRCG
XLRCG18   V18    V19  LRCG
XLRCG19   V19    V20  LRCG
ROUT      V20    0    50
*=========LRCG_MODEL=========================
.SUBCKT   LRCG  IN    OUT
 RS       IN    INS   1m
 RBP      IN    OUT   200
 L1       INS   OUT   12n
 C1       OUT   0     4.8p
 RP       OUT   0     1meg
.ENDS     LRCG
*                     _  _  _
*  ___          RS   / \/ \/ \      ___  3.75n/1cm
* |IN |__/\  /\  /\__| () () |_____|OUT| 1.5p/1cm
* |___|    \/  \/        L1    |   |___|           2120Mhz
*                      ________|       ___   ____   ___
*                RP   |       _|_     |IN |_|LRCG|_|OUT|
*        _/\  /\  /\__|       ___     |___| |____| |___|
*       _|_ \/  \/             |  C1          _|_
*       ///                   _|_             ///
*                             ///

.control
set              pensize = 2
echo             "===================Want_150_2ns_periods==============================="
let n =          150
let tstep =      1ns
let period_t =   n*tstep
echo             "Sample_Period_s =  $&tstep"
echo             "===================Create_arrays_=============="
unlet pwl_1
unlet noise
unlet ii
let pwl_1 =      vector(4*n)
let noise =      vector(n)
let ii =         vector(1*$&n)
echo             "===================create_Noise_array====================="
let index =      0
repeat           $&n
let              noise[index] = pos(rnd(127)-64)
let index =      index + 1
end
*plot            noise  vs ii
echo             "===================create_PWL_array====================="
                 pwl_1[0] = 0
                 pwl_1[1] = noise[0]
                 pwl_1[2] = tstep -1p
                 pwl_1[3] = noise[0]
let  n2          = n-1
let index =      1
repeat           $&n2
                 pwl_1[0+4*index] = pwl_1[4*index-4] +tstep
                 pwl_1[1+4*index] = noise[index]
                 pwl_1[2+4*index] = pwl_1[0+4*index] +tstep -1p
                 pwl_1[3+4*index] = noise[index]
let index =      index + 1
end
echo             "===================Install_the_PWL_array============================"
alter            @vin[pwl] = pwl_1
echo             "===================Run_and_Plot====================================="
let period_s =   tstep/10
*tran            0.5us 100us
*tran            $&period_s $&period_t   0  $&period_s
tran             .1n 150n   0  .1n
plot             vin vr vsaw
plot             vin/2 v10 v20
let npts =       length(time)-1
let m  =         0 
repeat           $&npts
if               ( abs(vsaw[m]) > .95)
let t = time[m]
let v10[m] =     0
let v20[m] =     0
endif
let m  =         m +1
end
plot             vin/2+.6 v10 v20-.7  vs vsaw
.endc



.end

